

Technical Supplementary Appendix

Yerushalmi, E., Hunt P., Hoorens S., Sauboin C., Smith R.

Exploring the use of a general equilibrium method to assess the value of a malaria vaccine: an application to Ghana

MDM Policy & Practice.

A Introduction

This is the technical appendix for the the paper “Exploring the use of a general equilibrium method to assess the value of a malaria vaccine: an application to Ghana ”. We develop a multi-sector multi-agent recursive dynamic CGE model, which is coded in GAMS¹ using the MPSGE high-level language by Rutherford (1995, 1999).² MPSGE allows handling of CGE models in a consistent and compact format.

B The general equilibrium model

We setup an Arrow–Debreu equilibrium as a mixed complementarity problem (Mathiesen, 1985; Rutherford, 1995, 1999). This enforces two variables to be complementary to each other; the following conditions hold for scalar variables x and y : $x \cdot y = 0$, $x \geq 0$, $y \geq 0$, which we compactly express as $0 \leq x \perp y \geq 0$.

The following is a summary of the main sets used in the model,

45 households	h
time	t
{self-employed, skilled, unskilled}	$\in l$ Labor
$\{K_{jt}, Lnd_{jt}, L_{ljt \in L}\}$	$\in f$ Factors
j	$\in m$ Activities
i	$\in n$ Commodities

¹For further information on GAMS, see www.gams.com.

²For further information about MPSGE, see <https://www.gams.com/solvers/mpsge/>.

B.1 Production

We use the Armington assumption, similar to Breisinger et al. (2009) and Hosoe et al. (2010), which is diagrammatically illustrated in Figure 1.

Inputs into production assemble six activities Y_j (i.e., four agricultural activities that represent differences in Ghana's four agro-ecological zones, industry and services). Using MPSGE, it is a simple matter to introduce substitution between the intermediate inputs in the production chain. However, at this high level of aggregation, this seems unnecessary since it would be very low in any case. Our results, therefore, might be more conservative in terms of the contribution that the intermediate inputs would have as a result of the malaria vaccine.

We use a standard Leontief-Cobb Douglas technology. Once solved, we obtain the demand for inputs and price indexes as in equations 3 through 7.

Firm j 's profit-maximization problem is the following: first

$$\begin{aligned} & \underset{Y_j, VA_j, A_{ij}}{\text{maximize}} && p_{Y,jt} Y_{jt} - \left(p_{VA,jt} VA_{jt} + \sum_i p_{it} A_{ijt} \right) \\ & \text{s.t.} && Y_{jt} \geq \min \left\{ \frac{VA_{jt}}{a_{jt}^{VA}}, \frac{A_{ijt}}{a_{ijt}^A}, \dots, \frac{A_{njt}}{a_{njt}^A} \right\}, \quad \forall j \end{aligned} \quad (1)$$

then


$$\begin{aligned} & \underset{VA_j, K_j, Lnd_j, L_{lj}}{\text{maximize}} && p_{VA,j} VA_j - \left(r_K K_{jt} + p_{Lnd} Lnd_j + \sum_{l=1}^3 w_{lt} L_{ljt} \right) \\ & \text{s.t.} && VA_j \geq \theta_j^{VA} K_{jt}^{\alpha_K} Lnd_{jt}^{\alpha_{Lnd}} \left(\prod_{l=1}^3 L_{ljt}^{\alpha_l} \right), \quad \forall j, \forall l \end{aligned} \quad (2)$$

with A_{ij} the intermediate inputs, VA_j a value added composite of factor inputs $\{K_j, Lnd_j, L_{lj}\} \in f$ capital, land and labor (respectively) that also includes three types of labor[1. self-employed, 2. skilled, 3. unskilled] $\in l$. p_j^Y is price of the activity output, p_j^{VA} the price of the value added composite, $w_{lj} \in p_f$ labor wage rates (with set f for factor), $p_j^{Lnd} \in p_f$ rental price of land, and $r_j^K \in p_f$ rental price of capital. From the Ghana Social Accounting Matrix (SAM) we calibrate the following: a_j^{VA}, a_{ij}^A the input requirement coefficients for the j 'th firm, α_{jf} share coefficient in the value added function, with constant returns to scale implying that $\alpha_k + \alpha_{Lnd} + \sum_{l=1}^3 \alpha_l = 1$. Finally, θ_j^{VA} is a scaling coefficient.

Solving these two maximization problems $\forall t, \forall j$, and $\forall i$ leads to the following demand for inputs:³

³These equations, as do the other equations in the the next sections, expresses the *complementarity* problem. Using equation (6) as an example for a *zero profit condition*, if $Y_j > 0$, the cost of inputs equals the revenue, hence zero profits, $p_{jt}^Y = a_{jt}^{VA} p_{jt}^{VA} + \sum_{i=1}^A a_{ijt}^A p_{ijt}^A$. However, if input costs are great than revenue, $p_{jt}^Y < a_{jt}^{VA} p_{jt}^{VA} + \sum_{i=1}^A a_{ijt}^A p_{ijt}^A$, a firm will not produce and $Y_j = 0$.

Figure 1: The production function



Note: Activity production inputs has a two level nested function. The lowest level combines capital, labour and land into an aggregate value added, VA_{jt} . The second level combines intermediate goods, A_{ijt} , with the value added to form output, Y_{jt} . Production output is then transformed into Export, E_{it} , and domestic consumption, D_{it} . Finally, domestic consumption and imports, M_{it} , are aggregated to form the Armington final good. This good is then demanded for private and public consumption, investment, or as an intermediate good.

$$0 \leq p_{jt}^{VA} \perp VA_{jt} \geq a_{jt}^{VA} Y_{jt} \quad (3)$$

$$0 \leq p_{jt}^A \perp A_{ijt} \geq a_{ijt}^A Y_{jt} \quad (4)$$

$$0 \leq p_{ft} \perp F_{jt} \geq \frac{\alpha_f p_{jt}^Y}{p_{ft}} Y_{jt} \quad \forall F \quad (5)$$

Zero profit conditions lead to the price indexes:

$$p_{jt}^Y \leq \left(a_{jt}^{VA} p_{jt}^{VA} + \sum_{i=1}^N a_{ijt}^A p_{ijt}^A \right) \perp Y_{jt} \geq 0 \quad (6)$$

$$p_{jt}^{VA} \leq \theta_j^{VA} \Pi_f \left\{ (p_{jt}^f)^{\alpha_{jf}} \right\} \perp VA_{jt} \geq 0 \quad (7)$$

B.1.1 Transformation between domestic use and exports

Activities Y_j are then aggregated into three main commodities Y_i , i.e., agricultural, industrial and service sectors as in equation 8. These are supplied to domestic markets D_i and foreign markets E_i , through a constant elasticity of transformation (CET) function as in equation 9, with elasticity of transformation $\eta = \frac{1}{\phi-1}$.⁴ The solution is the supply function for exports and domestic goods and their

⁴The transformation elasticity is defined by $\eta = \frac{d(D_i/E)}{(D_i/E)} / \frac{d(p_i^D/p_i^E)}{(p_i^D/p_i^E)}$

respect unit costs as in equations 10 to 14 (redundant equations were written for convenience).

We use a two stage process: first, activities $j \in m$ are assembled into a commodity $i \in n$ using a Leontief function

$$\begin{aligned} & \underset{Y_i, Y_j}{\text{maximize}} && p_{it}^Y Y_{it} - \sum_j^M p_{jt}^Y Y_{jt} \\ & \text{s.t.} && Y_{it} \geq \min \left\{ \frac{Y_{jt}}{a_{jt}^Y}, \dots, \frac{Y_{mt}}{a_{mt}^Y} \right\} \end{aligned} \quad (8)$$

which is then divided by the CET function

$$\begin{aligned} & \underset{Y_{it}, D_{it}, E_{it}}{\text{maximize}} && (p_{it}^D D_{it} + (1 - \tau_i^E) p_{it}^E E_{it}) - p_{it}^Y Y_{it} \\ & \text{s.t.} && Y_{it} \geq g(D_{it}, E_{it}) = \theta_{it} \left(\gamma_i^Y D_{it}^\phi + (1 - \gamma_i^Y) E_{it}^\phi \right)^{\frac{1}{\phi}} \end{aligned} \quad (9)$$

with p_i^D the price of the domestic output, p_i^E price of the exported good in terms of domestic currency with τ_i^E export tax. θ_i is a scaling coefficient, γ_i^Y share coefficients for good i 's transformation, calibrated from SAM.

Solving the maximization problem leads to the following supply functions for exports and domestic goods:

$$0 \leq p_{it}^Y \perp Y_{it} \geq a_{jt}^Y Y_{jt} \quad (10)$$

$$0 \leq p_i^E \perp E_{it} \geq \left(\theta_{it} \gamma_i^Y \frac{p_{it}^Y}{(1 - \tau_i^E) p_i^E} \right)^{\frac{1}{1-\phi}} Y_{it} \quad (11)$$

$$0 \leq p_i^D \perp D_{it} \geq \left(\theta_{it} (1 - \gamma_i^Y) \frac{p_{it}^Y}{p_i^D} \right)^{\frac{1}{1-\phi}} Y_{it} \quad (12)$$

The zero profit conditions lead to the unit cost of production of the total output,

$$p_{it}^Y \leq \sum_{j \in agr} a_{jt}^Y p_{jt}^Y \perp Y_{it} \geq 0 \quad (13)$$

$$p_{it}^Y \leq \left(\gamma_i^Y (p_{it}^D)^{1+\eta} + (1 - \gamma_i^Y) (p_{it}^E)^{1+\eta} \right)^{\frac{1}{1+\eta}} \perp Y_{it} \geq 0 \quad (14)$$

B.1.2 The Armington assumption

Finally, to account for cross-hauling (i.e., the export and import of the same good), we assume an Armington final good A_i is assembled by combining the domestic commodity D_i with imports M_i as imperfect substitutes, as in equation 15 (Armington, 1969). Thus, the demand function for imports and domestic goods are presented in equations 16 to 18.

The optimization problem for the i 'th final good

$$\begin{aligned} & \underset{A_{it}, M_{it}, D_{it}}{\text{maximize}} && (1 - \tau_i^S) p_{it}^A A_{it} - (1 + \tau_i^M) p_{it}^M M_{it} - p_{it}^D D_{it} \\ \text{s.t. } & A_{it} \geq g(M_{it}, D_{it}) &= & \theta_{it}^A \left(\gamma_i^A M_{it}^{\eta_i^A} + (1 - \gamma_i^A) D_{it}^{\eta_i^A} \right)^{\frac{1}{\eta_i^A}} \end{aligned} \quad (15)$$

with p_i^A consumer price of the Armington composite good, with the produce price being $(1 - \tau_i^S) p_i^A$. Furthermore, p_i^M is price of the imported good in terms of domestic currency, τ_i^S sales tax, τ_i^M import tariff, θ_i^A scaling coefficient, γ_i^A input share coefficients calibrated from SAM, η_i^A parameter defined by the substitution elasticity σ_i^A , having $\eta_i^A = 1 - \frac{1}{\sigma_i^A}$.

The first-order conditions for the optimality of the above problem $\forall t$ result in the following demand functions for imports and domestic goods:

$$0 \leq p_{it}^M \perp M_{it} \geq [\theta_i^A]^{\sigma_i^A - 1} \left(\gamma_i^A \frac{(1 - \tau_i^S) p_{it}^A}{(1 + \tau_i^M) p_{it}^M} \right)^{\sigma_i^A} A_{it} \quad (16)$$

$$0 \leq p_{it}^D \perp D_{it} \geq [\theta_i^A]^{\sigma_i^A - 1} \left((1 - \gamma_i^A) \frac{(1 - \tau_i^S) p_{it}^A}{p_{it}^D} \right)^{\sigma_i^A} A_{it} \quad (17)$$

$$p_{it}^A \leq \left(\gamma_i^A [(1 + \tau_i^M) p_{it}^M]^{1 - \sigma_i^A} + (1 - \gamma_i^A) (p_{it}^D)^{1 - \sigma_i^A} \right)^{\frac{1}{1 - \sigma_i^A}} \perp Y_{it} \geq 0 \quad (18)$$

B.2 Household behavior

Household h is endowed with an initial amount of labor, capital and land. Overtime, they accumulate (or lose) capital, transfer (or receive) funds from the government and the rest of the world, and pay taxes. As discussed in the main paper, effective labor endowments L_l are directly affected by the changes to malaria through demographic changes and labor efficiency changes. Furthermore, to com-

pare welfare across scenarios, it is assumed that the government's real level of services is fixed and rises proportionately with the population growth rate. Otherwise, it would be impossible to assess whether improvements in household welfare is due to efficiency improvements, or because the government is running a deficit. We therefore assume that households transfers (receives) an endogenous fraction of income ϕ_{ht} to meet this constraint. Overall, the disposable income is summarized by

$$\begin{aligned} Z_{ht} = & r_t^K K_{ht} + p_t^{Lnd} Lnd_{ht} + \sum_{l=1}^3 w_{lt} L_{lht} + pfx_t \cdot (HR_{ht} - RH_{ht}) \\ & + p_{gov,t} ((HG_{ht} - GH_{ht}) + \Phi_{ht} \cdot govtr_t - T_{ht}^d) - \sum_i p_{it}^A \bar{c}_{iht} \end{aligned} \quad (19)$$

with $\Phi_{ht} = \frac{C_{ht}}{\sum_h C_{ht}}$ so that $\sum_h \Phi = 1$, and $govtr$ the government deficit (or surplus). Furthermore, Lnd_h, p^{Lnd} are the land endowments and unit rental price, L_{lh}, w_l labor endowments and wage rates for {self, skilled, unskilled} $\in l$, and K_{ht}, r_t^K capital and rental price of capital. $(HG_{ht} - GH_{ht})$ are net household receipts from the government, and $(HR_{ht} - RH_{ht})$ is the net household receipts from the rest of world (ROW).

The household demand structure has two levels, and uses an extended linear expenditure system (ELES). First, a household consumes a composite consumption bundle, C_h , and saves a fixed share of disposable income, PSV_h .

$$\begin{aligned} \text{Maximize}_{S_{ht}, C_{ht}} \quad U_t = & \min \left\{ \frac{PSV_{ht}}{s_h^p}, \frac{C_{ht}}{1 - s_h^p} \right\}, \quad \text{with } 0 \leq s_h \leq 1 \\ \text{s.t.} \quad Z_{ht} \geq & p_t^{Inv} S_{ht} + p_{ht}^c C_h, \quad t \in T \end{aligned} \quad (20)$$

Second, the household maximizes a Stone-Geary utility function. The supernumerary income, M_h , is equal to the residual disposable income (net of taxes, transfers and savings).

$$\begin{aligned} \text{Maximize}_{c_{iht}} \quad C_{ht} (c_{iht} \cdots c_{nht}) = & \prod_{i=1}^n (c_{iht} - \bar{c}_{ih})^{\beta_{ih}} \quad \text{where } \sum_{i=1}^n \beta_{ih} = 1 \\ \text{s.t.} \quad M_{ht} \geq & \sum_{i=1}^n p_{it}^A c_{iht} \end{aligned} \quad (21)$$

with T_{ht}^D the direct tax by household h , PSV_h, p^{Inv} private savings and unit cost of investment, C_h, p_h^c demand for composite consumption and price index, Z_h disposable income (net subsistence level), τ_i^{sales} sales tax, c_{ih}, p_i^A demand for good i and consumer price, with \bar{c}_i the subsistence level. β_{hi} is the share

parameter for good i , with $0 \leq \beta_i \leq 1$ and $\sum_i \beta_i = 1$. p_{ht}^U is the unit cost of household utility.

Solving the household's problem, $\forall t$, the *first-level* demands for private savings and consumption bundle are:

$$0 \leq p_{ht}^c \perp C_{ht} \geq (1 - s_h^p) \frac{Z_{ht}}{p_{cht}} \quad (22)$$

$$0 \leq p_{inv,t} \perp PSV_t \geq s_h^p \frac{Z_{ht}}{p_{ht}^U} \quad (23)$$

with households utility price index of

$$p_{ht}^U \leq s_h^p p_t^{Inv} + (1 - s_h^p) p_{ht}^c \perp U_{ht} \geq 0 \quad (24)$$

In the *second-level*, the demand function for good i is

$$0 \leq p_{it} \perp c_{iht} \geq \bar{c}_{ih} + \beta_{ih} \frac{[M_{ht} - \sum_{j=1}^n p_{jt} \bar{c}_{jh}]}{p_{it}} \quad (25)$$

with household's consumption price index of

$$p_{ht}^c \leq \Phi^h \Pi_{i=1}^N \left\{ (p_{it}^A)^{\beta_{hi}} \right\} \perp C_{ht} \geq 0 \quad (26)$$

where $M_{ht} - \sum_i p_{it} \bar{c}_i$ is the supernumerary income and $c_{iht} - \bar{c}_{ih}$ is the supernumerary consumption of final good i , i.e., residual income net of subsistence expenditure, and residual consumption net of subsistence level.

B.3 Government behavior

The government receives income from collecting direct tax, τ^D , and sales tax, τ^S , including import and export tariffs, τ^M , τ^E (respectively). It also transfers (receives) funds from domestic households and the rest of the world. The government purchases commodities, and saves the remaining income. The level of real government services is assumed to be fixed, but rises proportionately with the population growth rate, *popgrowth* (i.e., fixed per-capita). As previously discussed, this constraint is maintained by receiving from (or transferring to) the households an endogenous lump-sum fund, *govdef*.

Government income, $\forall t$, is

$$\begin{aligned} GOVINC_t &= \sum_i T_{it}^S + \sum_i T_{it}^M + \sum_j T_{jt}^E + pfx_t \cdot (GR_t - RG_t) \\ &+ p_{gt} \cdot \left(\sum_h T_{ht}^D + \sum_h (GH_{ht} - HG_{ht}) + govtr_t \right) \quad \forall i, \forall j \end{aligned} \quad (27)$$

with tax revenues collected by sales tax, import and export tariffs $T_i^S = \tau_i^S p_{it}^A c_{iht}$, $T_i^M = \tau_i^M p_i^M M_i$, and $T_j^E = \tau_j^E p_j^E E_j$ for $\forall i$ and $\forall j$.

Government spends a fixed proportion of income on consumption and savings,

$$\begin{aligned} \underset{GSV_t, c_{git}}{\text{maximize}} G_t &= \min \left\{ \frac{GSV_t}{s_g}, \frac{c_{it}^g}{a_i^g}, \dots, \frac{c_{nt}^g}{a_N^g} \right\} \\ \text{s.t. } G_t &= G_{t=0} \cdot \text{popgrowth} \end{aligned} \quad (28)$$

with $0 \leq s_g, a_i^g \leq 1$ and $s_g + \sum_i a_i^g = 1$, characterizing a CRS function. T^D is the revenue from direct tax, $GOVINC$ government income, G, p_g level of government services and unit cost of government services, $(GH_{ht} - HG_{ht})$ net government receipts from households, $(GR_t - RG_t)$ net government receipts from the rest of the world, and $govdef$ total household transfers to cover government deficit. c_i^g, a_i^g are the government demand with fixed proportion, GSV, p^{Inv}, s_g government savings, unit cost of investment, and savings rate.

Therefore, $\forall t$, the government's demand for the i 'th good and savings are

$$0 \leq p_{it}^A \perp c_{it}^g \geq a_i^g \frac{GOVEXP_t}{p_{it}^A} \quad \forall i \quad (29)$$

$$0 \leq p_t^{Inv} \perp GSV_t \geq s_g \frac{GOVEXP_t}{p_t^{Inv}} \quad (30)$$

and the unit cost of government services and balanced budget constraint are

$$p_{gt} \leq s_g p_t^{Inv} + \sum_{i=1}^n a_{it}^g p_{it}^A \perp G_t \geq 0 \quad (31)$$

$$G_t = G_0 \cdot \text{popgrowth} \perp govtr \geq 0 \quad (32)$$

B.4 Rest of the world (ROW), international trade, and capital flow

We assume a small open economy that cannot affect world prices, and that export and import prices quoted in foreign currency are exogenously given.

The rest of the world (ROW) is modeled as a simple agent that demands foreign savings in the domestic economy. Its income is determined by

$$ROWINC_t = r_t^K K_{row,t} + pfx_t \cdot \left[\sum_h (RH_{ht} - HR_{ht}) + (RG_t - GR_t) \right] - NX_t^{base} \quad (33)$$

and the simple maximization problem is

$$\max \quad FSV_t \quad (34)$$

$$\text{s.t.} \quad ROWINC_t \geq p_{inv,t} FSV_t \quad (35)$$

where $ROWINC$ is the income of ROW, $\sum_h (RH_{ht} - HR_{ht})$ net total remittances from households, $(RG_t - GR_t)$ net remittances from the government, FSV foreign savings, NX^{base} an initially endowed net imports (a net exports from the perspective of domestic economy), K_{row} domestic capital owned by foreign agents. The demand function for foreign savings and unit cost for this activity are

$$0 \leq p_{row,t} \perp FSV_t \geq \frac{ROWINC_t}{p_{row,t}} \quad (36)$$

$$p_{row,t} \leq p_{inv,t} \perp FSV \geq 0 \quad (37)$$

ROW has an exogenous, baseline demand for net imports, NX^{base} (a net export from the point of view of the domestic economy). In the long run, trade must be balanced, but without additional foresight, the closure rule used is to increase the baseline level of net exports by the projected growth rate of the ROW working age population - that comes from the demographics model.

Export and import prices quoted in foreign currency are exogenously given by first, converting a unit of a foreign good, denominated in foreign exchange, into domestic prices

$$0 \leq M_{it} \perp \varepsilon_t \cdot pfx_t - p_{it}^M \geq 0, \quad \forall i \quad (38)$$

with pfx the unit price of the good in foreign currency. Second, a unit of domestic good is exchanged

(exported) for a unit of foreign currency by

$$0 \leq E_{it} \perp p_{it}^E - \varepsilon_t \cdot pfx_t \geq 0, \quad \forall i \quad (39)$$

and note that ε is the nominal exchange rate (domestic currency in terms of foreign currency), which is always assumed to be fixed (e.g. $\forall t, \varepsilon_t = 1$) and hence redundant.

B.5 Market clearing conditions, and trade balance closure rule

The market clearing conditions, $\forall t$, are

$$A_{it} = c_{iht} + c_{it}^g + c_i^{Inv} + \sum_j A_{ij} \quad \forall i, \forall j, \forall h \quad (40)$$

$$\sum_j L_{ljt} = \sum_h L_{lht}^s \quad \forall l, \forall h \quad (41)$$

$$\sum_j K_{jt} = \sum_h K_{ht}^s + K_{gov,t}^s + K_{ROW,t}^s \quad \forall j, \forall h \quad (42)$$

$$\sum_j Lnd_{jt} = \sum_h Lnd_{ht}^s \quad \forall j, \forall h \quad (43)$$

B.6 Investment and savings in the capital market

Investment I in new capital requires final goods in fixed proportion $0 \leq a_i^{Inv} \leq 1$, with $\sum_i a_i^{Inv} = 1$. The households and government's level of savings is determined by equations (30) and (23), with foreign savings in domestic currency determined by equation (36). Therefore, $\forall t$, total savings is

$$TSV_t = PSV_t + GSV_t + FSV_t \quad (44)$$

and investment is

$$\begin{aligned} \underset{c_{it}^{Inv}}{\text{maximize}} \quad I_t &= \min \left\{ \frac{c_{it}^{Inv}}{a_{it}^{Inv}}, \dots, \frac{c_{Nt}^{Inv}}{a_{Nt}^{Inv}} \right\} \\ \text{s.t.} \quad TSV_t &\geq \sum_i p_{it}^A c_{it}^{Inv} \end{aligned} \quad (45)$$

with c_i^{Inv} demand for investment of the i 'th Armington good, p_t^{Inv} unit cost of the investment good. The demand for investment is obtained by,

$$0 \leq p_{it}^A \perp c_{it}^{Inv} \geq a_{it}^{Inv} \frac{(PSV_t + GSV_t + FSV_t)}{p_{it}^A} \quad (46)$$

$$p_t^{Inv} \leq \sum_i a_{it}^{Inv} p_{it}^A \perp I_t \geq 0 \quad (47)$$

C Recursive Dynamics

We use the standard capital accumulation assumption of

$$K_{t+1} = (1 - \delta) K_t + I_t \quad (48)$$

with δ the depreciation rate. The one-year future price in terms of present value is given $p_{t+1} (1 + r) = p_t$.

At each period, capital has two types of prices: (i) the rental price of capital, r_t^K , and (ii) a unit purchase price of new capital, $p_{K,t}$. Assuming capital markets are competitive, the purchasing price of one unit of new capital equals the rental earnings of that unit, plus the value of the remaining capital sold in the subsequent period. The complementarity formulation of this problem is

$$0 \leq K_t \perp p_{K,t} \geq r_t^K + (1 - \delta) p_{K,t+1} \quad (49)$$

Furthermore, an agent decides between using goods for consumption or investment, and as in Section B.6, with p_t^{Inv} the unit cost of building an investment good. Assuming a fully dynamic model, the Euler condition equates the marginal utility of investment and capital accumulation $0 \leq I \perp p_t^{Inv} \geq p_{K,t+1}$. Since goods prices of two adjacent time periods are $p_t^{Inv} = (1 + r) p_{t+1}^{Inv}$, this implies that the capital purchase price equals $1 + r$ times the current cost of investment consumption,

$$(1 + r) p_t^{Inv} \geq p_{K,t} \quad (50)$$

Combining equation (50) with equation (49) leads to

$$r_t^K = (r + \delta) p_t^{Inv} \quad (51)$$

Normally, social accounting matrices do not supply capital stock, but rather the capital earnings

from services denoted by VK_t , which equals the capital stock, K_t , times the rental price of capital, r_t^K ,

$$VK_t = r_t^K \cdot K_t \quad (52)$$

Thus finally, using equation (51) with equation (52) into the capital accumulation equation (48) yields

$$\begin{aligned} VK_{t+1} &= \frac{1}{1+r} [(1-\delta)VK_t + (r+\delta)p_t^{Inv}I_t] \\ I_0 &\geq 0 \\ VK_0 &\geq 0 \end{aligned} \quad (53)$$

and initializing the model using VK_0 and I_0 from the SAM, and with $p_0^{Inv} = 1$ and $r_0^K = r + \delta$. Equation (53) is used for all agents in the model: h households, government, and the rest of the world. As discussed by Paltsev (2004), a forward-looking model would require that the initial values in the SAM, and especially investment, would be on a steady state growth path, so that the results in the final time period approximate the path of the infinite horizon. However, this is not necessary in our recursive model.

C.1 Updating stock variables

Without any better information, some stock variables are updated by the growth rate of the working age population g_{ht} that comes from the demographics-health component. For example, though land is in fixed supply, effective land inputs rise at per-working age, which is an implicit technology improvement in land-use.⁵ Others variables are updated by the growth rate of the working age population in the rest of the world row_t . Total factor productivity $TFP_{i,t}$ increases by g_t^{TFP} , which we obtained from economic studies on African and Ghana. Note however that since we only care about the marginal change between the baseline and the counterfactual malaria scenarios, the total factor productivity and the ROW working age growth rate play no role in our analysis on malaria because they are the same in all scenarios.

The following updating assumptions have been made:

⁵As discussed by Breisinger et al. (2011) for Ghana's case, agricultural growth in Ghana has been mainly driven by land productivity, which continues to expand at an annual rate of 2.8 percent. This is a bit higher than the average population growth rate that we use.

$$(HG_{h,t+1} - GH_{h,t+1}) = (HG_{ht} - GH_{ht}) \cdot (1 + g_{ht}) \quad (54)$$

$$(HR_{h,t+1} - RH_{h,t+1}) = (HR_{ht} - RH_{ht}) \cdot (1 + g_{ht}) \quad (55)$$

$$\bar{c}_{ih,t+1} = \bar{c}_{iht} \cdot (1 + g_{ht}) \quad (56)$$

$$T_{h,t+1}^d = T_{ht}^d \cdot (1 + g_{ht}) \quad (57)$$

$$Lnd_{h,t+1} = Lnd_{h,t} \cdot (1 + g_{ht}) \quad (58)$$

$$(GR_{t+1} - RG_{t+1}) = (GR_t - RG_t) \cdot (1 + row_t) \quad (59)$$

$$NX_{t+1}^{base} = NX_t^{base} \cdot (1 + row_t) \quad (60)$$

$$TFP_{i,t+1} = TFP_{it} \cdot (1 + g_t^{TFP}) \quad (61)$$

D The values used for the model parameters

To account for the subsistence level, \bar{c}_{ih} , the model is calibrated in the following steps: First, from the SAM, $\sum_{i=1}^n p_i^A c_{ih} = M_h$ and therefore, the average budget share is $s_i = \frac{p_i^A c_{ih}}{M_h}$. Second, the income elasticity of demand for good i is defined as $\epsilon_{i,h}^M = \frac{\partial c_{ih}}{\partial M_h} \frac{M_h}{c_{ih}}$. Differentiating the consumption demand Equation 25 with respect to disposable income, obtain $\frac{\partial c_{ih}}{\partial M_h} = \frac{\beta_{ih}}{p_i}$, and therefore, $\epsilon_{i,h}^M = \frac{\beta_{ih}}{p_i} \frac{M_h}{c_{ih}}$. Third, combine with the average budget share and rearrange, obtain the *marginal* budget share

$$\beta_{ih} = s_i \cdot \epsilon_{i,h}^M \quad (62)$$

where $\epsilon_{i,h}^M$ comes from econometric studies, and s_i is given from the SAM.

Finally, to calibrate the minimum subsistence requirement \bar{c}_{ih} , the values for the Frisch parameter are used from previous studies of African countries. The Frisch parameter is defined as $\phi_h = \frac{-M_h}{M_h - \sum_{j=1}^n p_j \bar{c}_{jh}}$, and reflects the marginal utility of income with respect to income, which tends to become smaller in absolute value as income rises. It measures the willingness of consumers to substitute between consumption of essential and non-essential goods (Frisch, 1959; Howe, 1975; De Melo and Tarr, 1992; Creedy, 1998; Lluch et al., 1977).

Placing the Frisch Equation into equation 25 rearranging to solve for \bar{c}_{ih} , obtain the calibrated subsistence level

$$\bar{c}_{ih} = c_{ih} + \frac{\beta_{ih}}{p_i} \cdot \frac{M_h}{\phi_h} \quad (63)$$

Finally, the benchmark social accounting matrix is revised so that a consumer is initially “endowed”

Table 1: The parameter values

Sectors	Income elasticity of demand ϵ_{ih}^M
Agricultural	0.67
Non Agricultural	1.2
Frisch parameter ϕ_h	-4
Total factor productivity (TFP) g_t^{TFP}	1.6%
Long-run interest rate	5%
Working age growth rate $g_{h,t}$	From the demographics model
ROW working age growth rate row_t	From the demographics model
Substitution elasticities	$\sigma_{1j}^Y = 0, \sigma_{2j}^Y = 1, \sigma_i^A = 4$
Transformation elasticities	$\eta_i = 2$

with \bar{c}_i , and the second-level composite utility function C_i is a Cobb-Douglas function with inputs of $c_{ih} - \bar{c}_{ih}$, with β_i re-scaled so that that $\sum \beta_i = 1$.

D.1 Parameters used in the model

We require income elasticities for the three aggregate sectors, i.e., Agricultural, Industrial and Services. We therefore collect income elasticities for various sectors from past econometric studies, and calculate a weighted average income elasticity based on the consumption shares of the specific sectors from the original, highly-disaggregative, Ghana SAM. [Nganou \(2005\)](#); [Hertel et al. \(2008\)](#); [Shimeles \(2010\)](#) estimate these for sub-Saharan Africa in general.⁶ Table 1 summarizes the main values used in this model.

For the Frisch parameter ϕ_h , we use -4, based on values by [Lluch et al. \(1977\)](#), [Hertel et al. \(1997\)](#) and [Nganou \(2005\)](#).

We do not actually require adding a total factor productivity (TFP) parameter into the model because our analysis relies on comparing counterfactual scenarios of malaria intervention to a baseline with *no* additional intervention. In other words, we are not interested in a forecast model, but rather to compare one scenario to another. However, for the sake of making the projected levels in the model more realistic, we use a total factor productivity of 1.6%, which is an approximate figure reported by [Arora and Bhundia \(2003\)](#); [Bezabih et al. \(2010\)](#), who had studied sub-Saharan Africa. Finally, the long-run interest rate is fixed to 5%, which is a value used in many applied general equilibrium papers.

⁶We use a weighted average because a simple average would be incorrect. For example, a sector with a high income elasticity but a low consumption share would lead to an overestimate of the aggregated income elasticity of demand.

E The GAMS code

Below is the GAMS code for the simulation. All data in excel format can be obtain upon request.

```
* Erez Yerushalmi - Erez.Yerushalmi@bcu.ac.uk

$title Malaria Recursive Model

* Stone-Geary utility
* The model's MPSGE code for GAMS

$ONTEXT
$MODEL:malaria

$SECTORS:
Y(a)      ! Output
XD(c)     ! diving export and domestic
ARM(c)    ! Armington
INV       ! Investment
CON(h)    ! Consumption
GOVPROD   ! Government services production
UTIL(h)   ! Welfare
UROW     ! Utility Rest of World
E(c)     ! Export
M(c)     ! Import

$COMMODITIES:
prow
pu(h)     ! utility level
py(a)$(sum(f,FA0(f,a))>0) ! output price
pc(h)     ! Consumption price
pgov      ! Government price index
pa(c)     ! Armington price index
pinv      ! Cost of investment
RK        ! Return to capital
PL(fl)$(sum(h,L0(h,fl))>0) ! Wage rate
plnd      ! return on land
PE(c)     ! domestic export price
PM(c)     ! domestic import price
pfx       ! foreign exchange
pd(c)     ! domestic index price

$CONSUMERS:
RA(h)     ! Representative agent
GOV       ! Government
ROW       ! Rest of World

$AUXILIARY:
GOVDEF    ! Household covers government deficit

$PROD:Y(a) s:0 kl(s):1 sl(s):0
o:py(a)   q:YTFP(a)
i:PA(c)   q:CA0(c,a)          sl:
i:PL(fl)  q:LD0(fl,a)        kl:
i:plnd    q:LND0(a)          kl:
i:RK      q:KD0(a)           kl:

$PROD:XD(c) t:2 va:0
o:PD(c)   q:DS0(c)
o:PE(c)   q:ER0(c) p:(1-t_e(c)) a:GOV t:t_e(c)
i:PY(a)   q:AC0(a,c)

$PROD:ARM(c) s:2
o:PA(c)   q:AS0(c) a:GOV t:t_s(c)
i:PD(c)   q:DS0(c)
i:PM(c)   q:RC0(c) p:(1+t_m(c)) a:GOV t:t_m(c)

$PROD:E(c) s:0
o:PFX     q:(ER0(c)*PWE0(c))
i:PE(c)   q:(ER0(c))

$PROD:M(c)
o:PM(c)   q:(RC0(c))
i:PFX     q:(RC0(c)*PWM0(c))

$PROD:INV
O:pinv    q:IT0
I:pa(c)   q:I0(c)
```

```

$PROD:CON(h) s:1
    O:pc(h) q:(CT_H0(h)-sum(c,sub0(c,h)))
    I:pa(c) q:(C0(c,h)-sub0(c,h))

$PROD:UTIL(h) s:0
    O:pu(h) q:(CT_H0(h)-sum(c,sub0(c,h))+PSV0(h))
    I:pc(h) q:(CT_H0(h)-sum(c,sub0(c,h)))
    I:pinv q:(PSV0(h))

$PROD:GOVPROD s:0
    o:pgov q:(GDT0+GSV0)
    i:pa(c) q:GD0(c)
    i:pinv q:(GSV0)

$PROD:UROW
    O:prow q:(FSV0)
    I:pinv q:(FSV0)

$DEMAND:RA(h)
    d:pu(h) q:(CT_H0(h)-sum(c,sub0(c,h))+PSV0(h))
*   d:pa(c) q:(sub(c,h))
    e:pa(c) q:(-sub(c,h))
    e:PL(fl) q:LS(h,fl)
    e:plnd q:LND(h)
    e:RK q:VK(h)
    e:PFX q:(HR(h)-RH(h)) ! Net Foreign remittances from ROW
    e:pgov q:(c_share(h) r:GOVDEF ! balance gov - lumpsum transfer of goverment saving.
    e:pgov q:(-sum(tx,TP(tx,h))) ! direct taxes to government
    e:pgov q:(HG(h)-GH(h))

$DEMAND:GOV
    d:pgov q:(GDT0+GSV0)
    e:pgov q:(sum(h,(sum(tx,TP(tx,h)))) ! taxes from households
    e:pgov q:l r:GOVDEF ! households cover government deficit
    e:pxf q:(GR-RG) ! transfers between government and ROW
    e:pgov q:(sum(h,(GH(h)-HG(h)))) ! direct gov transfers to households

$DEMAND:ROW
    d:Prow q:(FSV0)
    e:pxf q:(-NXb) ! Trade Balance
    e:PFX q:(sum(h,(RH(h)-HR(h)))) ! Net Foreign remittances from ROW
    e:pxf q:(RG-GR) ! transfers between government and ROW
    e:RK q:VKROW ! FDI Capital owned by ROW

$CONSTRAINT:GOVDEF
GOVPROD =E= 1*totpopgrow;

$REPORT:
    V:LD(fl,a) I:PL(fl) PROD:Y(a)
    V:KD(a) I:RK PROD:Y(a)
    V:Y_t(a) O:py(a) PROD:Y(a)
    V:INTERM(c,a) I:pa(c) PROD:Y(a)
    V:ITa O:pinv PROD:INV
    V:PSV(h) I:pinv PROD:UTIL(h)
    V:GSV I:pinv PROD:GOVPROD
    V:MD(c) I:pxf PROD:M(c)
    V:ED(c) O:pxf PROD:E(c)
    V:FDI I:RK PROD:SOE1
    V:DDI O:RK PROD:SOE2
    V:FSV I:pinv PROD:UROW
    V:Arming(c) O:pa(c) PROD:ARM(c)
    V:C_sub(h) I:pc(h) PROD:UTIL(h)
    V:GDT_GSV O:pgov PROD:GOVPROD
    V:GD(c) I:PA(c) PROD:GOVPROD
    V:UROW2 W:ROW
    V:INC(h) D:pu(h) DEMAND:RA(h)

$offtext
$sysinclude mpsgeset malaria
Y.L(a) = 1;
GOVDEF.LO=-INF;

malaria.ITERLIM = 200000;

$include malaria.GEN
SOLVE malaria USING MCP;
ABORT$(malaria.OBJVAL > 0.1) "Model does not calibrate.";

```


References

- Armington, P.: 1969, 'A Theory of Demand for Products Distinguished by Place of Production'. International Monetary Fund (IMF) Staff Papers 16.
- Arora, V. B. and A. Bhundia: 2003, 'Potential Output and Total Factor Productivity Growth in Post-Apartheid South Africa'. Working Paper 03/178, International Monetary Fund.
- Bezabih, M., M. Chambwera, and J. Stage: 2010, 'Climate Change, Total Factor Productivity, and the Tanzanian Economy: A Computable General Equilibrium Analysis'. Discussion Paper Series EFD DP 10-14, Environment for Development.
- Breisinger, C., X. Diao, and J. Thurlow: 2009, 'Modeling growth options and structural change to reach middle income country status: The case of Ghana'. *Economic Modelling* **26**(2), 514–525.
- Breisinger, C., X. Diao, J. Thurlow, and R. M. A. Hassan: 2011, 'Potential impacts of a green revolution in Africa—the case of Ghana'. *Journal of International Development* **23**(1), 82–102.
- Creedy, J.: 1998, 'Measuring the Welfare Effects of Price Changes: A Convenient Parametric Approach'. *Australian Economic Papers* **37**(2), 137–151.
- De Melo, J. and D. G. Tarr: 1992, *A General Equilibrium Analysis of US Foreign Trade Policy*. Cambridge, Mass. ; London: MIT Press.
- Frisch, R.: 1959, 'A Complete Scheme for Computing All Direct and Cross Demand Elasticities in a Model with Many Sectors'. *Econometrica* **27**(2), 177–196.
- Hertel, T., B. Dimaranan, and R. McDougall: 1997, 'GTAP 3 Data Base Documentation - Chapter 18'. http://www.gtap.agecon.purdue.edu/resources/res_display.asp?RecordID=846.
- Hertel, T., R. McDougall, B. Narayanan, and A. Aguiar: 2008, 'GTAP 7 Data Base Documentation - Chapter 14: Behavioral Parameters'. http://www.gtap.agecon.purdue.edu/resources/res_display.asp?RecordID=2937.
- Hosoe, N., K. Gasawa, and H. Hashimoto: 2010, *Textbook of Computable General Equilibrium Modelling: Programming and Simulations*. Palgrave Macmillan.
- Howe, H.: 1975, 'Development of the extended linear expenditure system from simple saving assumptions'. *European Economic Review* **6**(3), 305–310.

- Lluch, C., A. A. Powell, and R. A. Williams: 1977, *Patterns in Household Demand and Saving*. Oxford University Press.
- Mathiesen, L.: 1985, 'Computational Experience in Solving Equilibrium Models by a Sequence of Linear Complementarity Problems'. *Operations Research* **33**(6), 1225–1250.
- Nganou, J.: 2005, 'Estimation of the parameters of a linear expenditure system (LES) demand function for a small African economy'. MPRA Paper 31450, University Library of Munich, Germany.
- Paltsev, S.: 2004, 'Moving from Static to Dynamic General Equilibrium Economic Models (Notes for a beginner in MPSGE)'. Technical Note 4, MIT.
- Rutherford, T. F.: 1995, 'Extension of GAMS for complementarity problems arising in applied economic analysis'. *Journal of Economic Dynamics and Control* **19**(8), 1299–1324.
- Rutherford, T. F.: 1999, 'Applied General Equilibrium Modeling with MPSGE as a GAMS Subsystem: An Overview of the Modeling Framework and Syntax'. *Computational Economics* **14**, 1–46.
- Shimeles, A.: 2010, 'Welfare Analysis Using Data from the International Comparison Program for Africa'. Working Paper Series 122, African Development Bank Group.